

Cuprins

<i>Cuvânt înainte (Mihai Pricope)</i>	7
<i>Prefață</i>	9
<i>Mulțumiri</i>	11
Capitolul 1. Prima aplicație RIA	13
<i>Rich Internet Application</i>	13
De ce instrumente aveți nevoie?.....	15
Prima aplicație Flash	17
Redactarea codului AS3.....	23
Capitolul 2. Fundamentele limbajului ActionScript 3	27
Tipuri de date primitive (variabile, constante).....	27
Operatori.....	39
Conversia între tipurile de date.....	50
Instrucțiuni de bază AS3.....	51
Funcții definite de programator	59
Definirea unei clase de către programator	64
Tratarea erorilor.....	94
Testarea apartenenței și conversia la o anumită clasă	99
Concluzii.....	101
Capitolul 3. Pachetul de bază (Top Level)	103
Nucleul AS3.....	103
Utilizarea instanțelor clasei <i>Object</i> ;	104
Utilizarea șirurilor de caractere - <i>String</i> ;.....	106
Utilizarea tablourilor - <i>Array</i> ;.....	110
Utilizarea obiectelor de tip dată calendaristică - <i>Date</i> ;.....	117
Utilizarea funcțiilor matematice - <i>Math</i> ;	119
Utilizarea expresiilor regulate - <i>RegExp</i> ;	120
Clasele <i>int</i> ; <i>uint</i> ; <i>Number</i> ;	127
Utilizarea datelor în format XML - <i>XML</i> & <i>XMLList</i>	128
Alte clase din <i>Top Level</i>	146
Crearea și utilizarea unei biblioteci externe.....	152
Concluzii.....	156
Capitolul 4. Utilizarea evenimentelor	157
Modul de organizare a claselor	157

Pachetul <code>flash.events.*</code> ;	159
Concluzii	173
Capitolul 5. Elemente grafice	175
Grafica în AS3	175
Pachetul <code>flash.geom.*</code> ;	177
Pachetul <code>flash.display.*</code> ;	187
Concluzii	257
Capitolul 6. Utilizarea obiectelor textuale	259
Pachetul <code>flash.text</code> ;	259
<i>Text Layout Framework</i>	281
Capitolul 7. Utilizarea filtrelor	282
Pachetul <code>flash.filters.*</code> ;	282
<i>Pixel Bender</i>	297
Capitolul 8. Multimedia	299
Pachetul <code>flash.media.*</code> ;	299
Capitolul 9. Conectivitate în AS3	335
Preluarea datelor din documentul HTML	335
Comunicarea cu un server Web	339
Comunicarea între Flash și JavaScript	344
Comunicarea între două aplicații Flash	348
Utilizarea obiectelor de tip <i>SharedObject</i>	351
Comunicarea în timp real între două aplicații Flash aflate pe calculatoare diferite	353
Accesul la resurse aflate la distanță – politici de securitate	371
Includerea resurselor în aplicație	375
Consumarea serviciilor Web	378
Capitolul 10. Crearea unui joc în AS3	395
Realizarea unui joc de tip puzzle	395
<i>Bibliografie</i>	427

Metoda `appendChild` poate insera mai multe noduri simultan dacă primește ca parametru un obiect de tipul `XMLList` (evident, nodurile vor fi elementele listei).

Copilul nou adăugat obiectului `xmlObj` va fi așezat pe ultima poziție.

Pentru a adăuga copii pe primele poziții, în locul comenzii `appendChild` utilizați metoda `prependChild`.

Acestea nu sunt singurele posibilități. Pentru a insera noul nod după un anumit nod deja existent, se poate utiliza metoda `insertChildAfter` care are ca parametru două obiecte de tip XML: primul este nodul după care trebuie făcută inserarea, al doilea este nodul care trebuie inserat. De exemplu, pentru a insera nodul reprezentând noul prieten între cele două noduri deja existente în XML ar trebui ca, în locul metodei `appendChild`, să se utilizeze:

```
xmlObj.insertChildAfter( xmlObj.children()[0], prietenNou);
```

Metoda `insertChildBefore` are tot doi parametri și inserează obiectul reprezentat de cel de-al doilea parametru înaintea nodului identificat prin primul parametru.

Metoda cea mai simplă pentru inserarea unui nou nod este, totuși, prin adăugarea utilizând operatorul punct. Pentru aceasta, scrieți numele instanței obiectului, urmată de punct și de numele marcatorului ce va fi adăugat. Folosiți, apoi, operatorul egal pentru a atribui noului nod un obiect XML:

```
xmlObj.prieten = prietenNou;
```

Pentru a afla poziția pe care se află un anumit nod (de exemplu, pentru a o utiliza împreună cu comanda `children`), se poate apela metoda `childIndex`. Pentru exemplul de mai sus, se va afișa pe ce poziție se află nodul „data_nastere” în cadrul primului prieten existent în XML:

```
trace( xmlObj.children()[0].data_nastere.childIndex() );
```

Copierea unui obiect XML se va realiza cu metoda `copy`. Metoda `copy` nu necesită niciun argument, va returna un obiect XML identic cu cel care a apelat metoda, aflat în schimb la altă locație de memorie. Atunci când selectați un anumit nod și operați o modificare asupra nodului, modificarea se reflectă supra XML-ului original (încercați să selectați un prieten din lista de prieteni și să-i adăugați încă un prenume și veți vedea că modificarea se reflectă și în XML-ul original). Dacă utilizați metoda `copy`, pentru a copia în prealabil nodul într-un nou obiect, XML-ul rezultat va fi unul total nou și orice modificare făcută asupra sa nu va influența XML-ul din care a fost extras:

```
// vom clona primul nod într-un nou obiect XML:
var prieten : XML = xmlObj.children()[0].copy();

// clonei îi vom adăuga încă un element:
prieten.insertChildAfter( prieten.child("nume"),
                          <prenume>Cristi</prenume> );

// putem observa că originalul a rămas neschimbat:
trace( prieten.toXMLString() );
trace( "-----" );

// chiar dacă în clonă a apărut noul nume:
trace( xmlObj.toXMLString() );
```

Se poate observa că în cel de-al doilea element, primul nod de tip `prieten` din listă rămâne neschimbat.

Clasa `XML` nu furnizează o anumită metodă pentru eliminarea unui nod din arborele XML. Totuși, în AS3 există o metodă generică „delete” care elimină orice legătură dintre numele variabilei și locul din memorie în care este stocat obiectul. Obiectul în sine nu va mai putea referit și, din acest motiv, poate fi considerat șters. În momentul în care un obiect nu mai poate fi referit, el este trimis automat către *garbage-collector* (colectorul de gunoaie) care are rolul de a-l elimina din memorie (pentru a lăsa memoria liberă altor viitoare obiecte). Pentru a elimina primul nod din `xmlObj` putem să utilizăm o construcție de tipul:

```
delete xmlObj.children()[0];
trace( xmlObj );
```

Metoda `replace` are rolul de a înlocui un nod cu unul nou. Fie două obiecte: `xmlObj` și `prietenNou`. Metoda `replace` va fi apelată pentru obiectul în care se va face înlocuirea având ca parametri poziția în XML a nodului ce trebuie înlocuit și noul nod. De exemplu, dacă se dorește înlocuirea primului nod-`prieten` (identificat ca fiind pe poziția 0) cu `prietenNou`, se va scrie:

```
xmlObj.replace( 0 , prietenNou );
```

Atunci când obiectul XML conține mai multe noduri de tip text consecutive, ele pot fi concatenate într-un singur nod. De asemenea, un element care nu conține niciun fel de informație poate să nu fie scris ca `<element></element>` și direct `<element/>`.

Atunci când apare unul din cele două cazuri de mai sus, este bine ca XML-ul să fie normalizat (va ocupa mai puțin spațiu în memorie și interpretorul îl va putea parcurge mai ușor). Normalizarea unui obiect de tip XML se realizează cu ajutorul metodei `normalize`. Această metodă nu

are niciun parametru de intrare, nu returnează nimic, dar modifică în memorie obiectul XML apelant astfel încât să poată fi parcurs mai rapid. De exemplu:

```
var xml:XML =
    <test>
        <nod_vid></nod_vid>
    </test>

xml.appendChild("text1");
xml.appendChild("text2");
xml.appendChild("text3");
trace( xml );
trace("Înainte de normalizare:",xml.children().length());

xml.normalize();
trace( xml );
trace( "După normalizare: ", xml.children().length() );
```

Datorită adăugării nodurilor „text1”, „text2”, „text3”, înainte de normalizare, obiectul XML are patru copii (deși ultimii trei sunt doar noduri textuale, fără marcatori și nu ar trebui să se facă diferența între ei). După normalizare, cele trei noduri textuale sunt unite într-unul singur și elementul vid va fi afișat ca <nod_vid /> (de către `trace(xml);`).

Un conținut simplu este fie un nod textual, fie unul de tip element vid (chiar cu atribute). Un nod are conținut complex în toate celelalte cazuri. În cazul agendei, primul prieten (luat în ansamblu) formează un nod de tip complex, nodul care conține doar numele său este un nod de tip simplu. Pentru a testa dacă un nod are conținut simplu se va folosi `hasSimpleContent`, în mod identic se va apela `hasComplexContent` pentru a afla dacă un nod este complex. Cele două metode nu au niciun parametru și returnează o valoare logică în funcție de conținutul nodului apelant:

```
// pentru primul nod (prieten):
trace( xmlObj.children()[0].hasSimpleContent() ); //false
trace( xmlObj.children()[0].hasComplexContent() ); //true

// pentru nodul nume din primul nod prieten:
trace(xmlObj.children()[0].children()[0].
        hasSimpleContent()); // true
```

Atribute

Un atribut este o construcție de tipul `nume="valoare"` și poate fi adăugat în marcatorul de început al unui nod-element. Rolul unui atribut este de a

descrie proprietățile elementelor conținute de marcatorul în cauză. În cazul agendei, am adăugat, de fiecare dată, elementelor de tip prieten câte un atribut indicând sexul persoanei respective.

Pentru a prelua valorile tuturor atributelor existente într-un obiect de tip XML se va apela metoda `attributes` pentru respectivul obiect XML. Astfel, putem afișa valorile atributelor din primul nod prieten printr-o construcție:

```
trace( xmlObj.children()[0].attributes() );
```

Metoda `attributes` returnează un obiect de tip `XMLList`. Comanda `trace` apelează metoda `toString` pentru a putea afișa conținutul.

Pentru a prelua un anumit atribut (presupunând că sunt mai multe), se va utiliza metoda `attribute` care primește, ca argument, numele atributului și returnează valoarea acestuia. De exemplu:

```
trace( xmlObj.children()[0].attribute("sex") ); // m
```

O variantă și mai simplă de obținere a unui atribut este utilizarea operatorului `@`:

```
trace( xmlObj.children()[0].@sex ); // m
```

Pentru a adăuga un atribut unui marcator se va construi o referință către acel marcator plecând din rădăcina XML-ului, după care se va utiliza operatorul punct (utilizat și la adăugarea copiilor) urmat de caracterul `@` specific atributelor, numele atributului, operatorul de atribuire și valoarea atributului. În continuare, se va adăuga atributul „ani” primului prieten din agendă:

```
xmlObj.children()[0].@ani = 30;  
trace( xmlObj.children()[0].attribute("ani") ); // 30
```

Utilizarea spațiilor de nume

Din exemplele pe care le-am dat până în acest moment, se poate observa că o aplicație „se mulează” pe structura XML-ului pe care trebuie să-l utilizeze. Dacă dorim să construim un program care să utilizeze agenda, acest program trebuie să folosească corect semnificația marcatorului `prieten` sau a datei de naștere a acestuia.

Unele documente XML sunt standardizate, tocmai pentru ca fiecare aplicație ce ar încerca să le acceseze să cunoască un format clar, în care va găsi datele. Mai mult, pentru a indica că aceste tipuri de documente XML sunt particularizate, pentru a deservi anumitor scopuri, au fost redenumite (extensiile lor nu sunt XML, ci SVG, RSS etc.) și, în același timp, tipul lor

MIME¹ a fost schimbat (de exemplu: `application/xml+rss`). În acest mod, oricine va discuta despre SVG, de exemplu, va ști că acest format este special pentru reprezentarea elementelor grafice vectoriale (iar în cazul în care este un cunoscător al acestui format, ar putea enumera marcatorii ce pot fi utilizați în construcția unui SVG²). RSS este un alt format derivat din XML care permite sumarizarea conținutului unui sit (RSS = *Really Simple Syndication*). Acestea nu sunt singurele formate derivate din XML. Chiar și XML-ul din exemplul nostru (cel cu agenda) ar putea deveni un standard și toată lumea să știe în ce format trebuie să adauge informațiile în el pentru a fi compatibil cu aplicația noastră. De fapt, există un format de specificare a prietenilor utilizând documente XML cunoscut sub numele FOAF (*Friend Of A Friend*).

Nu este de ajuns să denumim altfel un fișier XML pentru a-l transforma într-un SVG sau RSS. O aplicație ar putea interpreta un anumit marcator dintr-un fișier SVG într-un anumit fel, în timp ce altă aplicație ar putea interpreta același marcator în alt fel. De exemplu, marcatorul „circle” din SVG ar putea fi interpretat de o aplicație realizată în Anglia ca un cerc, iar de alta realizată în Japonia ca fiind un grup de oameni. Pentru a fi siguri că toate aplicațiile „înțeleg” același marcaj în același fel și pentru a evita ambiguitățile, se utilizează un așa-zis spațiu de nume (specific fiecărui tip particular de document XML). Pentru un fișier în format SVG, acest spațiu de nume este specificat de adresa <http://www.w3.org/2000/svg>. La o astfel de adresă nu este neapărat să găsiți informații despre formatul respectiv (așa cum se întâmplă în cazul SVG-ului). Deoarece toată lumea folosește același spațiu de nume (care este unic pentru SVG), SVG-ul este recunoscut de toată lumea în același fel. Un obiect de tip XML poate fi inserat într-un document SVG. Pentru a nu se confunda tag-urile din HTML cu cele din SVG, se poate defini spațiul de nume caracteristic SVG și preciza această definiție de fiecare dată când este utilizat un marcator specific SVG. De exemplu (salvați într-un fișier cu extensia `svg`):

```
<html:html xmlns:html="http://www.w3.org/1999/xhtml"
           xmlns:svg="http://www.w3.org/2000/svg">
<html:body>
<html:p>Acesta este un paragraf HML</html:p>
  <svg:svg width="300" height="100" version="1.1">
```

1. Tipul MIME (*Multipurpose Internet Mail Extensions*) a fost utilizat pentru a identifica corect formatele de fișiere din cadrul unui e-mail. În prezent a fost extins și la alte protocoale (HTTP, RTP). Detalii la adresa: http://en.wikipedia.org/wiki/MIME_type.
2. Mai multe despre SVG la: <http://www.w3.org/Graphics/SVG/>.

```

    <svg:circle cx="100" cy="50" r="40" stroke="black"
              stroke-width="2" fill="red" />
  </svg:svg>
</html:body>
</html:html>

```

În acest exemplu, putem identifica câteva marcaje specifice HTML: `html`, `body`, `p` și câteva marcaje specifice SVG-ului: `svg`, `circle`. Pentru a nu se confunda între ele (pentru că nu fac parte din același limbaj), s-au definit spațiile de nume pentru HTML și pentru SVG (*xmlns* provine de la *XML Name Space*) și de fiecare dată când se utilizează un marcator (specific SVG-ului spre exemplu), acesta va fi prefixat de spațiul de nume corespunzător (*svg*) și de caracterul două puncte. Pentru a putea vizualiza¹ elementele grafice definite în exemplul de mai sus, veți avea nevoie să instalați un *player* de SVG: <http://www.adobe.com/svg/viewer/install/>.

Pentru a concluziona, numele de spații identifică tipul datelor care sunt conținute în documentul XML. Spațiul de nume poate fi definit ca un atribut din spațiul de nume *xmlns* care are o anumită valoare - un URI unde este definit spațiul de nume.

AS3 permite includerea spațiilor de nume prin intermediul unui obiect special numit *Namespace* (care se află, de asemenea, în „*Top Level*”, deci nu trebuie importată nici o clasă specială). Singurele chestiuni importante dintr-un obiect de tip *Namespace* sunt cele două proprietăți ale sale: `prefix` și `uri`, ambele de tip *String*. Proprietatea `prefix` este cea care indică cu ce se va identifica spațiul de nume (de exemplu, „*svg*” - cel definit în *xmlns* în exemplul de mai sus), proprietatea `uri` va conține adresa la care se află definiția spațiului de nume.

Exemplu de declarare și inițializare a unui obiect de tip *Namespace*:

```

var xmlns:Namespace = new Namespace("soap",
    "http://www.w3.org/2001/12/soap-envelope");

```

Cu ajutorul comenzii `setNamespace` se poate asocia un spațiu de nume unui anumit obiect XML:

```

var xmlns:Namespace = new Namespace("soap",
    "http://www.w3.org/2001/12/soap-envelope");

var xml:XML =
    <Envelope>
      <Body/>
    </Envelope>;

```

1. Unele browsere (ca, de exemplu, Firefox) au preinstalat un *player* (interpretor) de SVG.